In this worksheet we study the parametrizability of Maxwell equations. See H. Weyl, *Space Time Matter*, fourth edition, Dover, 1952.

```
> with(Ore_algebra):
> with(OreModules):
```

We define the Weyl algebra $Alg = A_4$, where d_i acts as differential operator w.r.t. x_i, i=1, ..., 4, and where x_1, x_2, x_3 are the spatial variables and x_4 is the time variable.

```
> Alg := DefineOreAlgebra(diff=[d[1],x[1]], diff=[d[2],x[2]], diff=[d[3],x[3]],
> diff=[d[4],x[4]], polynom=[x[1],x[2],x[3],x[4]], comm=[epsilon,mu]):
```

We enter the system matrix of the *first set of Maxwell equations*, which is a matrix with entries in *Alg*: The first three rows stand for the sum of the time derivative of the magnetic field and the curl of the *electric field*; the last row of the matrix is the divergence of the *magnetic field*.

```
> R := evalm([[d[4], 0, 0, 0, -d[3], d[2]],
> [0, d[4], 0, d[3], 0, -d[1]],
> [0, 0, d[4],-d[2], d[1], 0],
> [d[1], d[2], d[3], 0, 0, 0]]);
```

$$R := \begin{bmatrix} d_4 & 0 & 0 & 0 & -d_3 & d_2 \\ 0 & d_4 & 0 & d_3 & 0 & -d_1 \\ 0 & 0 & d_4 & -d_2 & d_1 & 0 \\ d_1 & d_2 & d_3 & 0 & 0 & 0 \end{bmatrix}$$

In terms of equations, the first set of Maxwell equations is given by:

```
> ApplyMatrix(R,
> [seq(B[i](seq(x[j],j=1..4)),i=1..3),seq(E[i](seq(x[j],j=1..4)),i=1..3)], Alg)
> =evalm([[0]$4]);
```

$$\begin{bmatrix} (\frac{\partial}{\partial x_4}\, B_1(x_1,\, x_2,\, x_3,\, x_4)) - (\frac{\partial}{\partial x_3}\, E_2(x_1,\, x_2,\, x_3,\, x_4)) + (\frac{\partial}{\partial x_2}\, E_3(x_1,\, x_2,\, x_3,\, x_4)) \\ (\frac{\partial}{\partial x_4}\, B_2(x_1,\, x_2,\, x_3,\, x_4)) + (\frac{\partial}{\partial x_3}\, E_1(x_1,\, x_2,\, x_3,\, x_4)) - (\frac{\partial}{\partial x_1}\, E_3(x_1,\, x_2,\, x_3,\, x_4)) \\ (\frac{\partial}{\partial x_4}\, B_3(x_1,\, x_2,\, x_3,\, x_4)) - (\frac{\partial}{\partial x_2}\, E_1(x_1,\, x_2,\, x_3,\, x_4)) + (\frac{\partial}{\partial x_1}\, E_2(x_1,\, x_2,\, x_3,\, x_4)) \\ (\frac{\partial}{\partial x_1}\, B_1(x_1,\, x_2,\, x_3,\, x_4)) + (\frac{\partial}{\partial x_2}\, B_2(x_1,\, x_2,\, x_3,\, x_4)) + (\frac{\partial}{\partial x_3}\, B_3(x_1,\, x_2,\, x_3,\, x_4)) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

The syzygy module of the *Alg*-module generated by the rows of R is defined by:

```
> Syzygy := SyzygyModule(R, Alg);
```

$$Syzygy := \begin{bmatrix} d_1 & d_2 & d_3 & -d_4 \end{bmatrix}$$

Equivalently, if we consider the first set of Maxwell equations with a right member $(\kappa[1] : ... : \kappa[4])^T$, then the compatibility condition of the inhomogeneous system is given by:

```
> ApplyMatrix(Syzygy, [seq(kappa[i](seq(x[j],j=1..4)),i=1..4)], Alg)[1,1]=0;
```

$$(\frac{\partial}{\partial x_1}\, \kappa_1(x_1,\, x_2,\, x_3,\, x_4)) + (\frac{\partial}{\partial x_2}\, \kappa_2(x_1,\, x_2,\, x_3,\, x_4)) + (\frac{\partial}{\partial x_3}\, \kappa_3(x_1,\, x_2,\, x_3,\, x_4)) \\ - (\frac{\partial}{\partial x_4}\, \kappa_4(x_1,\, x_2,\, x_3,\, x_4)) = 0$$

Let us check whether or not the first set of Maxwell equations is parametrizable. In order to do that, let us introduce the formal adjoint $R\_\ adj$ of R:

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -d_4 & 0 & 0 & -d_1 \\ 0 & -d_4 & 0 & -d_2 \\ 0 & 0 & -d_4 & -d_3 \\ 0 & -d_3 & d_2 & 0 \\ d_3 & 0 & -d_1 & 0 \\ -d_2 & d_1 & 0 & 0 \end{bmatrix}$$

To check whether the system of Maxwell equations is parametrizable, we compute the first extension module extˆ1 with values in *Alg* of the left *Alg*-module N which is associated with *R_adj*:

```
>  st := time():  Ext1 := Exti(R_adj, Alg, 1); time()-st;
```

$$Ext1 := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} d_4 & 0 & 0 & 0 & -d_3 & d_2 \\ d_1 & d_2 & d_3 & 0 & 0 & 0 \\ 0 & -d_4 & 0 & -d_3 & 0 & d_1 \\ 0 & 0 & d_4 & -d_2 & d_1 & 0 \end{bmatrix}, \begin{bmatrix} d_3 & d_2 & 0 & 0 \\ 0 & -d_1 & d_3 & 0 \\ -d_1 & 0 & -d_2 & 0 \\ 0 & 0 & -d_4 & -d_1 \\ d_4 & 0 & 0 & -d_2 \\ 0 & -d_4 & 0 & -d_3 \end{bmatrix} \right]$$

$$0.680$$

Since *Ext1*[1] is the identity matrix, we see that the module M, which is associated with R, is torsion-free. Equivalently, the system of Maxwell equations is parametrizable and *Ext1*[3] is a parametrization of the system. More precisely, if we introduce the following notations

```
>  Bvec := evalm([seq([B[i](seq(x[j],j=1..4))],i=1..3)]);
```

$$Bvec := \begin{bmatrix} B_1(x_1, x_2, x_3, x_4) \\ B_2(x_1, x_2, x_3, x_4) \\ B_3(x_1, x_2, x_3, x_4) \end{bmatrix}$$

```
>  Evec := evalm([seq([E[i](seq(x[j],j=1..4))],i=1..3)]);
```

$$Evec := \begin{bmatrix} E_1(x_1, x_2, x_3, x_4) \\ E_2(x_1, x_2, x_3, x_4) \\ E_3(x_1, x_2, x_3, x_4) \end{bmatrix}$$

then, we obtain the following classical parametrization of the first set of the Maxwell equation by means of the *quadri-potential* (A[1], A[2], A[3], V):

```
>  ApplyMatrix(Ext1[3],
>  [-A[2](seq(x[i],i=1..4)),A[3](seq(x[i],i=1..4)),A[1](seq(x[i],i=1..4)),
>  V(seq(x[i],i=1..4))], Alg)=linalg[stackmatrix](Bvec, Evec);
```

$$\begin{bmatrix} -(\frac{\partial}{\partial x_3} A_2(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_2} A_3(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_1} A_3(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_3} A_1(x_1, x_2, x_3, x_4)) \\ (\frac{\partial}{\partial x_1} A_2(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_2} A_1(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_4} A_1(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_1} \mathrm{V}(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_4} A_2(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_2} \mathrm{V}(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_4} A_3(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_3} \mathrm{V}(x_1, x_2, x_3, x_4)) \end{bmatrix} = \begin{bmatrix} B_1(x_1, x_2, x_3, x_4) \\ B_2(x_1, x_2, x_3, x_4) \\ B_3(x_1, x_2, x_3, x_4) \\ E_1(x_1, x_2, x_3, x_4) \\ E_2(x_1, x_2, x_3, x_4) \\ E_3(x_1, x_2, x_3, x_4) \end{bmatrix}$$

Let us point out that the parametrization of the first set of Maxwell equations is not unique. In particular, we had to permute some columns of *Ext1*[3] in order to find again the standard parametrization of the first set of Maxwell by the quadri-potential (A[1], A[2], A[3], V). In the rest of this Maple worksheet, in order to find again the classical results of electromagnetism, we shall use the following standard parametrization P instead of *Ext1*[3]:

```
>  grad := evalm([[d[1]], [d[2]], [d[3]]]):
```

```
> curl := evalm([[0, -d[3], d[2]], [d[3], 0, -d[1]], [-d[2], d[1], 0]]):
> P1 := linalg[stackmatrix](curl, linalg[band]([-d[4]], 3)):
> P2 := linalg[stackmatrix]([[0]$3], -grad):
> P := linalg[augment](P1, P2);
```

$$P := \begin{bmatrix} 0 & -d_3 & d_2 & 0 \\ d_3 & 0 & -d_1 & 0 \\ -d_2 & d_1 & 0 & 0 \\ -d_4 & 0 & 0 & -d_1 \\ 0 & -d_4 & 0 & -d_2 \\ 0 & 0 & -d_4 & -d_3 \end{bmatrix}$$

Let us insist that P is obtained by multiplying the second column by -1 and by performing a cyclic permutation on the first three colums of *Ext1*[3].

Therefore, P and *Ext1*[3] define the same parametrization up to a change of names of the arbitrary functions, i.e., of the potentials. We have:

```
> ApplyMatrix(P, [seq(A[i](seq(x[i],i=1..4)),i=1..3),V(seq(x[i],i=1..4))], Alg)
> =linalg[stackmatrix](Bvec, Evec);
```

$$\begin{bmatrix} -(\frac{\partial}{\partial x_3} A_2(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_2} A_3(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_1} A_3(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_3} A_1(x_1, x_2, x_3, x_4)) \\ (\frac{\partial}{\partial x_1} A_2(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_2} A_1(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_4} A_1(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_1} \mathrm{V}(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_4} A_2(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_2} \mathrm{V}(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_4} A_3(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_3} \mathrm{V}(x_1, x_2, x_3, x_4)) \end{bmatrix} = \begin{bmatrix} B_1(x_1, x_2, x_3, x_4) \\ B_2(x_1, x_2, x_3, x_4) \\ B_3(x_1, x_2, x_3, x_4) \\ E_1(x_1, x_2, x_3, x_4) \\ E_2(x_1, x_2, x_3, x_4) \\ E_3(x_1, x_2, x_3, x_4) \end{bmatrix}$$

In what follows, we shall see that this parametrization is not minimal. We first compute a free resolution of the *Alg*-module M associated with R:

```
> FreeResolution(R, Alg);
```

$$\mathrm{table}([1 = \begin{bmatrix} d_4 & 0 & 0 & 0 & -d_3 & d_2 \\ 0 & d_4 & 0 & d_3 & 0 & -d_1 \\ 0 & 0 & d_4 & -d_2 & d_1 & 0 \\ d_1 & d_2 & d_3 & 0 & 0 & 0 \end{bmatrix}, 2 = \begin{bmatrix} d_1 & d_2 & d_3 & -d_4 \end{bmatrix}, 3 = \mathrm{INJ}(1)])$$

In particular, by summing alternatingly the number of columns of all the entries in this free resolution, we find that the rank of M is 6 - 4 + 1 = 3. This result can also be obtained using *OreRank*:

```
> OreRank(R, Alg);
```

$$3$$

Hence, a minimal parametrization of the system involves only three potentials contrary to the parametrization by the quadri-potential. Let us compute some minimal parametrizations of the system using *MinimalParametrizations*:

```
> st := time(): Pmin := MinimalParametrizations(R, Alg); time()-st;
```

$$Pmin := \left[\begin{bmatrix} d_3 & d_2 & 0 \\ 0 & -d_1 & d_3 \\ -d_1 & 0 & -d_2 \\ 0 & 0 & -d_4 \\ d_4 & 0 & 0 \\ 0 & -d_4 & 0 \end{bmatrix}, \begin{bmatrix} d_3 & d_2 & 0 \\ 0 & -d_1 & 0 \\ -d_1 & 0 & 0 \\ 0 & 0 & -d_1 \\ d_4 & 0 & -d_2 \\ 0 & -d_4 & -d_3 \end{bmatrix}, \begin{bmatrix} d_3 & 0 & 0 \\ 0 & d_3 & 0 \\ -d_1 & -d_2 & 0 \\ 0 & -d_4 & -d_1 \\ d_4 & 0 & -d_2 \\ 0 & 0 & -d_3 \end{bmatrix}, \begin{bmatrix} d_2 & 0 & 0 \\ -d_1 & d_3 & 0 \\ 0 & -d_2 & 0 \\ 0 & -d_4 & -d_1 \\ 0 & 0 & -d_2 \\ -d_4 & 0 & -d_3 \end{bmatrix}\right]$$

0.690

We write the first of these minimal parametrizations in a more familiar way using the free functions ψ_1, ψ_2, ψ_3:

```
> ApplyMatrix(Pmin[1], [seq(psi[i](seq(x[j],j=1..4)),i=1..3)], Alg)
> =linalg[stackmatrix](Bvec, Evec);
```

$$\begin{bmatrix} (\frac{\partial}{\partial x_3}\psi_1(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_2}\psi_2(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_1}\psi_2(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_3}\psi_3(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_1}\psi_1(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_2}\psi_3(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_4}\psi_3(x_1, x_2, x_3, x_4)) \\ \frac{\partial}{\partial x_4}\psi_1(x_1, x_2, x_3, x_4) \\ -(\frac{\partial}{\partial x_4}\psi_2(x_1, x_2, x_3, x_4)) \end{bmatrix} = \begin{bmatrix} B_1(x_1, x_2, x_3, x_4) \\ B_2(x_1, x_2, x_3, x_4) \\ B_3(x_1, x_2, x_3, x_4) \\ E_1(x_1, x_2, x_3, x_4) \\ E_2(x_1, x_2, x_3, x_4) \\ E_3(x_1, x_2, x_3, x_4) \end{bmatrix}$$

Similar minimal parametrizations of the first set of Maxwell equations can be obtained using *Pmin*[2] and *Pmin*[3]. We finish by studying the second and the third extension modules with values in *Alg* of the *Alg*-module N associated with R_adj:

```
> st := time(): Ext2 := Exti(Involution(P, Alg), Alg, 1); time()-st;
```

$$Ext2 := \left[\begin{bmatrix} 1&0&0&0&0&0\\0&1&0&0&0&0\\0&0&1&0&0&0\\0&0&0&1&0&0\\0&0&0&0&1&0\\0&0&0&0&0&1 \end{bmatrix}, \begin{bmatrix} d_4&0&0&d_1\\-d_3&0&d_1&0\\-d_2&d_1&0&0\\0&d_4&0&d_2\\0&-d_3&d_2&0\\0&0&d_4&d_3 \end{bmatrix}, \begin{bmatrix} -d_1\\-d_2\\-d_3\\d_4 \end{bmatrix}\right]$$

0.880

Since *Ext2*[1] is the identity matrix, we see that ext^2 of N is zero, and thus, the *Alg*-module associated with the first set of Maxwell equations is reflexive.

Therefore, if we consider the system of equations formed by the parametrization of the first set of Maxwell equations by means of the quadri-potential ($A[1]$, $A[2]$, $A[3]$, V) with a zero right hand side, then this new system is parametrizable. In fact, such a parametrization is given in *Ext2*[3].

```
> ApplyMatrix(Ext2[3], [-xi(seq(x[i],i=1..4))], Alg)
> =evalm([seq([A[i](seq(x[i],i=1..4))],i=1..4)]);
```

$$\begin{bmatrix} \frac{\partial}{\partial x_1}\%1 \\ \frac{\partial}{\partial x_2}\%1 \\ \frac{\partial}{\partial x_3}\%1 \\ -(\frac{\partial}{\partial x_4}\%1) \end{bmatrix} = \begin{bmatrix} A_1(x_1, x_2, x_3, x_4) \\ A_2(x_1, x_2, x_3, x_4) \\ A_3(x_1, x_2, x_3, x_4) \\ A_4(x_1, x_2, x_3, x_4) \end{bmatrix}$$

$$\%1 := \xi(x_1, x_2, x_3, x_4)$$

This parametrization corresponds to the *gauge transformation* of the quadri-potential ($A[1]$, $A[2]$, $A[3]$, V). Obviously, this parametrization is a minimal one.

```
> st := time(): Ext3 := Exti(Involution(P,Alg), Alg, 2); time()-st;
```

$$Ext3 := \left[\begin{bmatrix} d_4\\d_3\\d_2\\d_1 \end{bmatrix}, \begin{bmatrix} 1 \end{bmatrix}, \mathrm{SURJ}(1)\right]$$

0.330

Since *Ext3*[1] is not an identity matrix, we see that extˆ3 of N is not the zero module. Therefore, the *Alg*-module associated with the first set of Maxwell equations is not projective, and thus, free. In particular, we cannot parametrize the above gauge transformation if $(A[1],\, A[2],\, A[3],\, V)^T = (0 : 0 : 0 : 0)^T$.

Let us interpret the formal adjoints of P, *Ext2*[3] and R. Let us start with the formal adjoint *P_ adj* of the parametrization P of the first set of Maxwell equations. We have:

```
> P_adj := Involution(P, Alg);
```

$$P\_adj := \begin{bmatrix} 0 & -d_3 & d_2 & d_4 & 0 & 0 \\ d_3 & 0 & -d_1 & 0 & d_4 & 0 \\ -d_2 & d_1 & 0 & 0 & 0 & d_4 \\ 0 & 0 & 0 & d_1 & d_2 & d_3 \end{bmatrix}$$

We easily check that *P_ adj* corresponds to the *second set of Maxwell equations*. More precisely, we have

```
> ApplyMatrix(P_adj, [seq(H[j](seq(x[j],j=1..4)),j=1..3),-D1(seq(x[j],j=1..4)),
> -D2(seq(x[j],j=1..4)),-D3(seq(x[j],j=1..4))], Alg)
> =evalm([seq([j[i](seq(x[j],j=1..4))],i=1..3),[-rho(seq(x[j],j=1..4))]]);
```

$$\begin{bmatrix} -(\frac{\partial}{\partial x_3} H_2(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_2} H_3(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_4} \mathrm{D1}(x_1, x_2, x_3, x_4)) \\ (\frac{\partial}{\partial x_3} H_1(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_1} H_3(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_4} \mathrm{D2}(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_2} H_1(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_1} H_2(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_4} \mathrm{D3}(x_1, x_2, x_3, x_4)) \\ -(\frac{\partial}{\partial x_1} \mathrm{D1}(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_2} \mathrm{D2}(x_1, x_2, x_3, x_4)) - (\frac{\partial}{\partial x_3} \mathrm{D3}(x_1, x_2, x_3, x_4)) \end{bmatrix} =$$

$$\begin{bmatrix} j_1(x_1, x_2, x_3, x_4) \\ j_2(x_1, x_2, x_3, x_4) \\ j_3(x_1, x_2, x_3, x_4) \\ -\rho(x_1, x_2, x_3, x_4) \end{bmatrix}$$

where $(H[1],\, H[2],\, H[3])$ denotes the *magnetic induction*, (D1, D2, D3) the *electric induction*, $(j[1],\, j[2],\, j[3])$ the *density of current* and ρ the *density of electric charge*. Now, the formal adjoint *R3_ adj* of the gauge transformation *Ext2*[3] is defined by:

```
> R3_adj := Involution(Ext2[3], Alg);
```

$$R3\_adj := \begin{bmatrix} d_1 & d_2 & d_3 & -d_4 \end{bmatrix}$$

We easily check that *R3_ adj* corresponds to the *conservation law*, as we have:

```
> ApplyMatrix(R3_adj, [seq(j[i](seq(x[j],j=1..4)),i=1..3),-rho(seq(x[j],j=1..4))],
> Alg)[1,1]=0;
```

$$(\frac{\partial}{\partial x_1} j_1(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_2} j_2(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_3} j_3(x_1, x_2, x_3, x_4)) + (\frac{\partial}{\partial x_4} \rho(x_1, x_2, x_3, x_4)) = 0$$

Finally, if we take $j[\mathrm{i}] = 0$ and $\rho = 0$, we can check whether or not the resulting second set of Maxwell equations are parametrizable.

```
> ext1 := Exti(P, Alg, 1);
```

$$ext1 := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} -d_3 & 0 & d_1 & 0 & -d_4 & 0 \\ -d_2 & d_1 & 0 & 0 & 0 & d_4 \\ 0 & -d_3 & d_2 & d_4 & 0 & 0 \\ 0 & 0 & 0 & d_1 & d_2 & d_3 \end{bmatrix}, \begin{bmatrix} -d_4 & -d_1 & 0 & 0 \\ 0 & -d_2 & d_4 & 0 \\ 0 & -d_3 & 0 & -d_4 \\ 0 & 0 & d_3 & d_2 \\ d_3 & 0 & 0 & -d_1 \\ -d_2 & 0 & -d_1 & 0 \end{bmatrix} \right]$$

Therefore, the second set of Maxwell equation is parametrized by *ext1*[3], i.e., we have the following parametrization of the magnetic/electric inductions:

```
> Hvec := evalm([seq([H[i](seq(x[j],j=1..4))],i=1..3)]):
> Dvec := evalm([[D1(seq(x[j],j=1..4))],[D2(seq(x[j],j=1..4))],[D3(seq(x[j],j=1..4))]]):
> ApplyMatrix(ext1[3], [seq(theta[i](seq(x[j],j=1..4)),i=1..4)], Alg)
> =linalg[stackmatrix](Hvec, Dvec);
```

$$\begin{bmatrix} -(\frac{\partial}{\partial x_4}\%2)-(\frac{\partial}{\partial x_1}\%4) \\ -(\frac{\partial}{\partial x_2}\%4)+(\frac{\partial}{\partial x_4}\%1) \\ -(\frac{\partial}{\partial x_3}\%4)-(\frac{\partial}{\partial x_4}\%3) \\ (\frac{\partial}{\partial x_3}\%1)+(\frac{\partial}{\partial x_2}\%3) \\ (\frac{\partial}{\partial x_3}\%2)-(\frac{\partial}{\partial x_1}\%3) \\ -(\frac{\partial}{\partial x_2}\%2)-(\frac{\partial}{\partial x_1}\%1) \end{bmatrix} = \begin{bmatrix} H_1(x_1,\,x_2,\,x_3,\,x_4) \\ H_2(x_1,\,x_2,\,x_3,\,x_4) \\ H_3(x_1,\,x_2,\,x_3,\,x_4) \\ \mathrm{D1}(x_1,\,x_2,\,x_3,\,x_4) \\ \mathrm{D2}(x_1,\,x_2,\,x_3,\,x_4) \\ \mathrm{D3}(x_1,\,x_2,\,x_3,\,x_4) \end{bmatrix}$$

$$\%1 := \theta_3(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%2 := \theta_1(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%3 := \theta_4(x_1,\,x_2,\,x_3,\,x_4)$$
$$\%4 := \theta_2(x_1,\,x_2,\,x_3,\,x_4)$$

The arbitrary functions θ[i], i=1, ..., 4, are sometimes called *pseudo-potentials*. Let us point out that the parametrization *ext1*[3] of the second set of Maxwell equations by the pseudo-potentials is equivalent to the formal adjoint *R_adj* of the first set of Maxwell equations:

Let us introduce the following matrix, where μ denotes the magnetic constant and ε is the dielectric constant:

```
> M := linalg[diag](evalm([[1/mu,0,0],[0,1/mu,0],[0,0,1/mu]]),evalm([[epsilon,0,0],
> [0,epsilon,0],[0,0,epsilon]]));
```

$$M := \begin{bmatrix} \frac{1}{\mu} & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{1}{\mu} & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{\mu} & 0 & 0 & 0 \\ 0 & 0 & 0 & \varepsilon & 0 & 0 \\ 0 & 0 & 0 & 0 & \varepsilon & 0 \\ 0 & 0 & 0 & 0 & 0 & \varepsilon \end{bmatrix}$$

Then, the *Minkowski law* is defined by:

```
> linalg[stackmatrix](Hvec, Dvec)=ApplyMatrix(M,
> [seq(B[i](seq(x[j],j=1..4)),i=1..3),seq(E[i](seq(x[j],j=1..4)),i=1..3)], Alg);
```

$$\begin{bmatrix} H_1(x_1,\,x_2,\,x_3,\,x_4) \\ H_2(x_1,\,x_2,\,x_3,\,x_4) \\ H_3(x_1,\,x_2,\,x_3,\,x_4) \\ \mathrm{D1}(x_1,\,x_2,\,x_3,\,x_4) \\ \mathrm{D2}(x_1,\,x_2,\,x_3,\,x_4) \\ \mathrm{D3}(x_1,\,x_2,\,x_3,\,x_4) \end{bmatrix} = \begin{bmatrix} \frac{B_1(x_1,\,x_2,\,x_3,\,x_4)}{\mu} \\ \frac{B_2(x_1,\,x_2,\,x_3,\,x_4)}{\mu} \\ \frac{B_3(x_1,\,x_2,\,x_3,\,x_4)}{\mu} \\ \varepsilon\,E_1(x_1,\,x_2,\,x_3,\,x_4) \\ \varepsilon\,E_2(x_1,\,x_2,\,x_3,\,x_4) \\ \varepsilon\,E_3(x_1,\,x_2,\,x_3,\,x_4) \end{bmatrix}$$

Now, if we substitute the parametrization of the first set of Maxwell equations by means of the quadri-potential ($A[1]$, $A[2]$, $A[3]$, V) into the Minkowski law and substitute again the result into the second set of Maxwell equations, we obtain the matrix of differential operators

```
> J := linalg[diag](linalg[band]([1], 3), linalg[band]([-1], 3)):
> wave := Mult(P_adj, J, M, P, Alg);
```

$$wave := \begin{bmatrix} \frac{-{d_3}^2 - {d_2}^2 + {d_4}^2\,\varepsilon\,\mu}{\mu} & \frac{d_2\,d_1}{\mu} & \frac{d_3\,d_1}{\mu} & d_4\,\varepsilon\,d_1 \\ \frac{d_2\,d_1}{\mu} & \frac{-{d_3}^2 - {d_1}^2 + {d_4}^2\,\varepsilon\,\mu}{\mu} & \frac{d_3\,d_2}{\mu} & d_4\,\varepsilon\,d_2 \\ \frac{d_3\,d_1}{\mu} & \frac{d_3\,d_2}{\mu} & \frac{-{d_2}^2 - {d_1}^2 + {d_4}^2\,\varepsilon\,\mu}{\mu} & d_4\,\varepsilon\,d_3 \\ d_4\,\varepsilon\,d_1 & d_4\,\varepsilon\,d_2 & d_4\,\varepsilon\,d_3 & {d_1}^2\,\varepsilon + {d_2}^2\,\varepsilon + {d_3}^2\,\varepsilon \end{bmatrix}$$

or, in terms of the ($A[1]$, $A[2]$, $A[3]$, V), we obtain the following equations:

```
> Equations := ApplyMatrix(wave, [seq(A[j](seq(x[i],i=1..4)),j=1..3),
> V(seq(x[i],i=1..4))], Alg):
> E1 := simplify(Equations[1,1])=j[1](seq(x[i],i=1..4));
```

$$\begin{aligned} E1 := {} & (\varepsilon\,(\tfrac{\partial^2}{\partial {x_4}^2}\,A_1(x_1, x_2, x_3, x_4))\,\mu - (\tfrac{\partial^2}{\partial {x_2}^2}\,A_1(x_1, x_2, x_3, x_4)) - (\tfrac{\partial^2}{\partial {x_3}^2}\,A_1(x_1, x_2, x_3, x_4)) \\ & + (\tfrac{\partial^2}{\partial x_2\,\partial x_1}\,A_2(x_1, x_2, x_3, x_4)) + (\tfrac{\partial^2}{\partial x_3\,\partial x_1}\,A_3(x_1, x_2, x_3, x_4)) \\ & + \varepsilon\,(\tfrac{\partial^2}{\partial x_4\,\partial x_1}\,\mathrm{V}(x_1, x_2, x_3, x_4))\,\mu)/\mu = j_1(x_1, x_2, x_3, x_4) \end{aligned}$$

```
> E2 := simplify(Equations[2,1])=j[2](seq(x[i],i=1..4));
```

$$\begin{aligned} E2 := {} & ((\tfrac{\partial^2}{\partial x_2\,\partial x_1}\,A_1(x_1, x_2, x_3, x_4)) + \varepsilon\,(\tfrac{\partial^2}{\partial {x_4}^2}\,A_2(x_1, x_2, x_3, x_4))\,\mu - (\tfrac{\partial^2}{\partial {x_1}^2}\,A_2(x_1, x_2, x_3, x_4)) \\ & - (\tfrac{\partial^2}{\partial {x_3}^2}\,A_2(x_1, x_2, x_3, x_4)) + (\tfrac{\partial^2}{\partial x_3\,\partial x_2}\,A_3(x_1, x_2, x_3, x_4)) \\ & + \varepsilon\,(\tfrac{\partial^2}{\partial x_4\,\partial x_2}\,\mathrm{V}(x_1, x_2, x_3, x_4))\,\mu)/\mu = j_2(x_1, x_2, x_3, x_4) \end{aligned}$$

```
> E3 := simplify(Equations[3,1])=j[3](seq(x[i],i=1..4));
```

$$\begin{aligned} E3 := {} & -(-(\tfrac{\partial^2}{\partial x_3\,\partial x_1}\,A_1(x_1, x_2, x_3, x_4)) - (\tfrac{\partial^2}{\partial x_3\,\partial x_2}\,A_2(x_1, x_2, x_3, x_4)) \\ & - \varepsilon\,(\tfrac{\partial^2}{\partial {x_4}^2}\,A_3(x_1, x_2, x_3, x_4))\,\mu + (\tfrac{\partial^2}{\partial {x_1}^2}\,A_3(x_1, x_2, x_3, x_4)) + (\tfrac{\partial^2}{\partial {x_2}^2}\,A_3(x_1, x_2, x_3, x_4)) \\ & - \varepsilon\,(\tfrac{\partial^2}{\partial x_4\,\partial x_3}\,\mathrm{V}(x_1, x_2, x_3, x_4))\,\mu)/\mu = j_3(x_1, x_2, x_3, x_4) \end{aligned}$$

```
> E4 := simplify(Equations[4,1])=-rho(seq(x[i],i=1..4));
```

$$\begin{aligned} E4 := {} & \varepsilon((\tfrac{\partial^2}{\partial x_4\,\partial x_1}\,A_1(x_1, x_2, x_3, x_4)) + (\tfrac{\partial^2}{\partial x_4\,\partial x_2}\,A_2(x_1, x_2, x_3, x_4)) + (\tfrac{\partial^2}{\partial x_4\,\partial x_3}\,A_3(x_1, x_2, x_3, x_4)) \\ & + (\tfrac{\partial^2}{\partial {x_1}^2}\,\mathrm{V}(x_1, x_2, x_3, x_4)) + (\tfrac{\partial^2}{\partial {x_2}^2}\,\mathrm{V}(x_1, x_2, x_3, x_4)) + (\tfrac{\partial^2}{\partial {x_3}^2}\,\mathrm{V}(x_1, x_2, x_3, x_4))) = \\ & -\rho(x_1, x_2, x_3, x_4) \end{aligned}$$

However, we saw that the quadri-potential ($A[1]$, $A[2]$, $A[3]$, V) is defined up to a gauge transformation. Therefore, we can assume that the quadri-potential ($A[1]$, $A[2]$, $A[3]$, V) satisfies the following equation:

```
> G := evalm([[seq(d[i],i=1..3),epsilon*mu*d[4]]]):
> ApplyMatrix(G, [seq(A[j](seq(x[i],i=1..4)),j=1..3),V(seq(x[i],i=1..4))], Alg)[1,1]=0;
```

$$\begin{aligned} & (\tfrac{\partial}{\partial x_1}\,A_1(x_1, x_2, x_3, x_4)) + (\tfrac{\partial}{\partial x_2}\,A_2(x_1, x_2, x_3, x_4)) + (\tfrac{\partial}{\partial x_3}\,A_3(x_1, x_2, x_3, x_4)) \\ & + \varepsilon\,\mu\,(\tfrac{\partial}{\partial x_4}\,\mathrm{V}(x_1, x_2, x_3, x_4)) = 0 \end{aligned}$$

Multiplying the previous equation by the following column of differential operators

```
> T := evalm([seq([-1/mu*d[i]], i=1..3),[-epsilon*d[4]]]);
```

$$T := \begin{bmatrix} -\frac{d_1}{\mu} \\ -\frac{d_2}{\mu} \\ -\frac{d_3}{\mu} \\ -\varepsilon\, d_4 \end{bmatrix}$$

we obtain the so-called *gauge condition* defined by *Gauge* $(A[1], A[2], A[3], V)^T = (0 : 0 : 0 : 0)^T$, where *Gauge* is the following matrix:

```
> Gauge := Mult(T, G, Alg);
```

$$Gauge := \begin{bmatrix} -\frac{{d_1}^2}{\mu} & -\frac{d_2\, d_1}{\mu} & -\frac{d_3\, d_1}{\mu} & -d_4\, \varepsilon\, d_1 \\ -\frac{d_2\, d_1}{\mu} & -\frac{{d_2}^2}{\mu} & -\frac{d_3\, d_2}{\mu} & -d_4\, \varepsilon\, d_2 \\ -\frac{d_3\, d_1}{\mu} & -\frac{d_3\, d_2}{\mu} & -\frac{{d_3}^2}{\mu} & -d_4\, \varepsilon\, d_3 \\ -d_4\, \varepsilon\, d_1 & -d_4\, \varepsilon\, d_2 & -d_4\, \varepsilon\, d_3 & -\varepsilon^2\, {d_4}^2\, \mu \end{bmatrix}$$

Then, from the systems *wave* $(A[1], A[2], A[3], V)^T = (j[1] : j[2] : j[3] : \text{-}\ \rho)^T$ and *Gauge* $(A[1], A[2], A[3], V)^T = (0 : 0 : 0 : 0)^T$, we finally obtain *Wave* $(A[1], A[2], A[3], V)^T = (j[1] : j[2] : j[3] : \text{-}\ \rho)^T$, where the matrix *Wave* is defined by:

```
> Wave := simplify(evalm(wave+Gauge));
```

$$Wave := \begin{bmatrix} \%1 & 0 & 0 & 0 \\ 0 & \%1 & 0 & 0 \\ 0 & 0 & \%1 & 0 \\ 0 & 0 & 0 & {d_1}^2\, \varepsilon + {d_2}^2\, \varepsilon + {d_3}^2\, \varepsilon - \varepsilon^2\, {d_4}^2\, \mu \end{bmatrix}$$

$$\%1 := \frac{-{d_2}^2 - {d_1}^2 + {d_4}^2\, \varepsilon\, \mu - {d_3}^2}{\mu}$$

Using the fact that $v^2 = 1/(\mu\, \varepsilon)$, where v is the speed of light, we see that the quadri-potential $(A[1], A[2], A[3], V)$ satisfies the following equations:

```
> WaveEq := ApplyMatrix(Wave, [seq(A[j](seq(x[i],i=1..4)),j=1..3),
> V(seq(x[i],i=1..4))], Alg):
> subs(epsilon=1/(mu*v^2), simplify(mu*WaveEq[1,1]))=mu*j[1](seq(x[i],i=1..4));
```

$$\frac{\frac{\partial^2}{\partial {x_4}^2}\, \%1}{v^2} - (\tfrac{\partial^2}{\partial {x_1}^2}\, \%1) - (\tfrac{\partial^2}{\partial {x_2}^2}\, \%1) - (\tfrac{\partial^2}{\partial {x_3}^2}\, \%1) = \mu\, j_1(x_1,\, x_2,\, x_3,\, x_4)$$
$$\%1 := A_1(x_1,\, x_2,\, x_3,\, x_4)$$

```
> subs(epsilon=1/(mu*v^2), simplify(mu*WaveEq[2,1]))=mu*j[2](seq(x[i],i=1..4));
```

$$\frac{\frac{\partial^2}{\partial {x_4}^2}\, \%1}{v^2} - (\tfrac{\partial^2}{\partial {x_1}^2}\, \%1) - (\tfrac{\partial^2}{\partial {x_2}^2}\, \%1) - (\tfrac{\partial^2}{\partial {x_3}^2}\, \%1) = \mu\, j_2(x_1,\, x_2,\, x_3,\, x_4)$$
$$\%1 := A_2(x_1,\, x_2,\, x_3,\, x_4)$$

```
> subs(epsilon=1/(mu*v^2), simplify(mu*WaveEq[3,1]))=mu*j[3](seq(x[i],i=1..4));
```

$$\frac{\frac{\partial^2}{\partial x_4{}^2}\,\%1}{v^2} - \left(\frac{\partial^2}{\partial x_1{}^2}\,\%1\right) - \left(\frac{\partial^2}{\partial x_2{}^2}\,\%1\right) - \left(\frac{\partial^2}{\partial x_3{}^2}\,\%1\right) = \mu\, j_3(x_1,\, x_2,\, x_3,\, x_4)$$

$$\%1 := A_3(x_1,\, x_2,\, x_3,\, x_4)$$

```
> subs(epsilon=1/(mu*v^2), simplify(-WaveEq[4,1]/epsilon))
> =epsilon*rho(seq(x[i],i=1..4));
```

$$\frac{\frac{\partial^2}{\partial x_4{}^2}\,\%1}{v^2} - \left(\frac{\partial^2}{\partial x_1{}^2}\,\%1\right) - \left(\frac{\partial^2}{\partial x_2{}^2}\,\%1\right) - \left(\frac{\partial^2}{\partial x_3{}^2}\,\%1\right) = \varepsilon\, \rho(x_1,\, x_2,\, x_3,\, x_4)$$

$$\%1 := \mathrm{V}(x_1,\, x_2,\, x_3,\, x_4)$$

Finally, if there is no densities of current and of electric charge, i.e., $j[\,i] = 0$, i=1, ..., 3, and $\rho = 0$, then we conclude that the quadri-potential $(A[1], A[2], A[3], V)$ satisfies an *electromagnetic wave* with a speed of propagation equals to the speed of light in the vacuum, i.e., $v = c = 1/\ (\varepsilon 0\, \mu 0)$, where $\mu 0$ and $\varepsilon 0$ are respectively the magnetic and the electric constants in the vacuum.